Stored Procedure

CREATE PROCEDURE [dbo].[MySP] 
	-- Add the parameters for the stored procedure here
        @p_country nvarchar(100)
AS
BEGIN
	-- SET NOCOUNT ON added to prevent extra result sets from
	-- interfering with SELECT statements.
	SET NOCOUNT ON;

    IF @p_country = ''
		SELECT * FROM Customers
	ELSE
		SELECT * FROM Customers Where Country = @p_country
END

Layout Customizations

<asp:dropdownlist id="Country" runat="server"/>


<asp:GridView ID="MyGrid" runat="server" 	OnPageIndexChanging="gridView_PageIndexChanging" 	allowpaging="True" pageindex="1"> 
	<RowStyle CssClass="ttc"></RowStyle> 
	<SelectedRowStyle CssClass="ttcs"></SelectedRowStyle> 
	<PagerStyle CssClass="ttc"></PagerStyle> 
	<HeaderStyle CssClass="thc"></HeaderStyle> 
</asp:GridView>


Code Customizations

Protected Sub Country_Init(ByVal sender As Object, ByVal e As System.EventArgs) Handles Country.Init
    Me.Country.AutoPostBack = True
End Sub


Protected Sub Country_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Country.Load
    Dim rec As CustomersRecord

    ' If already loaded, no need to load it again from the database
    If Me.Country.Items.Count > 0 Then Return

    Dim ob As OrderBy = New OrderBy(False, True)
    ob.Add(CustomersTable.Country, Asc)

    Dim item As ListItem
    For Each rec In CustomersTable.GetRecords("", ob)
        item = New ListItem(rec.Country, rec.Country)
        ' If the item is already in the list, do not add it.
        If Not Me.Country.Items.Contains(item) Then
            Me.Country.Items.Add(item)
        End If
    Next

    item = New ListItem("Please Select", "")
    Me.Country.Items.Insert(0, item)
End Sub


Protected Sub Country_SelectedIndexChanged(ByVal sender As Object, ByVal e As System.EventArgs) Handles Country.SelectedIndexChanged
    ' User changed country, so reload data using stored procedure
    LoadGridControl()
End Sub

Private Sub LoadGridControl()
    ' Enable pagination for the GridView control.
    Me.MyGrid.AllowPaging = True

    ' Get the data for GridView control from stored procedure.
    Me.MyGrid.DataSource = Me.CallCustomStoredProcedureForGridView()

    ' Save the data obtained from stored procedure in session to be used 
    ' re-bind the GridView control after postback occurs since GridView 
    ' control loses data after postback occurs.
    Me.Page.Session("DataSet") = Me.MyGrid.DataSource

    ' Let the GridView control handle rendering of columns and their names.
    Me.MyGrid.AutoGenerateColumns = True
    ' Bind the GridView control with the data from stored procedure.
    Me.MyGrid.DataBind()
    Me.MyGrid.Visible = True
End Sub


Protected Sub gridView_PageIndexChanging(ByVal sender As Object, ByVal e As GridViewPageEventArgs)
    ' Set the GridView control to display new page
    Me.MyGrid.PageIndex = e.NewPageIndex

    ' Obtain the stored procedure data that was saved in session.
    Dim ds As Object = Me.Page.Session("DataSet")
    If (Not IsNothing(ds)) Then
        ' Re-bind the GridView control with the data
        ' from session.
        Me.MyGrid.DataSource = CType(ds, DataSet)
        Me.MyGrid.DataBind()
        Me.MyGrid.Visible = True
    Else
        Me.MyGrid.Visible = False
    End If
End Sub


Private Function CallCustomStoredProcedureForGridView() As System.Data.DataSet
    '' Step 1: Configure parameters for your stored
    Dim firstParameter As BaseClasses.Data.StoredProcedureParameter
    firstParameter = New BaseClasses.Data.StoredProcedureParameter("@p_country", Me.Country.SelectedValue, _
                                System.Data.SqlDbType.VarChar, _
                                System.Data.ParameterDirection.Input)

    '' Step 2: Add the configured parameters to an array list.
    Dim parameterList(0) As BaseClasses.Data.StoredProcedureParameter
    parameterList(0) = firstParameter

    '' Skip steps 1 and 2 if stored procedure does not accept parameters.

    Dim myStoredProcedure As BaseClasses.Data.StoredProcedure

    '' "DatabaseNorthwind1" is a connection string obtained from Web.config 
    '' located in application's root directory.

    '' Step 3: Connect to the stored procedure using following line if 
    '‘ the stored procedure accepts parameters, use the following
    myStoredProcedure = New BaseClasses.Data.StoredProcedure( _
                                  "DatabaseNorthwind1", _
                                 "MySP", _
                                 parameterList)

    '' Step 4: Run the stored procedure.    
    '' RunQuery() will return true if stored procedure successfully executed.
    '' Otherwise, it will return false.  Use RunNonQuery for scalar results.
    If (myStoredProcedure.RunQuery()) Then
        '' Result from stored procedure is available 
        '' as a DataSet or as an array of RecordValue objects.
        Return (myStoredProcedure.DataSet)
    Else
        Return Nothing
    End If
End Function
